

# 퍼스날 컴퓨터에 의한 dobby 직물의 織造무늬 디자인

安 東 汶\* · 楊 激 坤

成均館大學校 工科大學 織維工學科

## Personal Computer Aided Design for Patterns of Dobby Woven Fabrics

Dong Moon Ahn · Cheol Kon Yang

Dept. of Textile Engineering, College of Engineering, Sung Kyun Kwan University

(1985. 6. 18 접수)

### Abstract

The purpose of this study is to develop a new method for the visual presentation of weaves and woven patterns of dobby fabrics by using the personal computer briefly.

The following results were given through adapting this method to the design for patterns of woven fabrics.

1. Complete weaves were given by computerizing the pegging plan and drafting system.
2. The effect of the combined woven patterns were given by computerizing the one complete weaves, the differently coloured warp threads, and the differently coloured weft threads.

### I. 緒 論

織物の外觀을 變化시키는 要因에는 經糸, 緯糸의 굵기로 부터, 織物の 經, 緯糸密度, 色糸의 配列 및 組織等 여러가지가 있다. 이러한 織物外觀의 變化를 製織하거나 그려보지 않고 위와 같은 資料를 컴퓨터에 入力하면 즉시, 正確하게 나타낼 수 있도록, 컴퓨터를 利用하려는 研究와 試圖는 날로 더해가고 있으며, 近年 外國에서 製織用 電子柄出 system 이 開發되어 圖案으로부터 畫像處理 system 에 의해 自動적으로 紋紙가 作成되어 Jacquard 機로 製織하게 되었다고 하며<sup>1)</sup>, 이것은 原圖를 몇개의 카메라로 入力후, 무늬처리 system 으로 무늬의 修正, 合成, 配色등을 畫像處理 소프트웨어로 무늬를 完成시키는 것이며, dobby 機에 관한 것으로는, 最近에 마이크로 컴퓨터를 利用하

여 dobby 무늬의 設計와 色糸配列에 의한 織造무늬를 나타내는 것이 開發되었다고 하며<sup>2)</sup>, 프랑스의 Pujos 社가 APPLE II 혹은 APPLE II.型을 使用하여 소프트웨어 프로그램을 開發했다고 하나<sup>3)</sup> 아직 그 內容이 公開되지 않고 간단히 소개만 되고 있는 실정이다. 따라서 本 研究에서는 우리 주변에서 흔히 접할 수 있는 퍼스날컴퓨터를 利用하여 이러한 프로그램을 자체개발하여 직물디자인과 섬유기계에서의 활용하는 초기 단계로서 도비직물의 디자인 및 設計, 나아가서는 織機에 응용해 보고자 하였다.

dobby 직물은 經糸의 通入方法 즉, 通經圖와 紋控圖에 의하여 織物組織이 形成되며, 紋控植法이나 綜統通入方法을 바꾸면 여러가지 組織으로 바뀌게 된다. 그러므로 이 두가지의 조건을 資料로 하여 資料入力 즉시 正確하게 모니터와 프린터에 나타나게 하는 방법과 또한 織物組織과 經糸, 緯糸의 색갈이나 配列을 변화시키면 여러가지 多樣한 무늬가 나타나는 프로그래밍

\* 부천공업전문대학 섬유공학과




Fig. 1. 완전의장도 그리는 과정

방법을 檢討하고 그 結果를 提示하여 織物디자인에 컴퓨터의 구체적인 利用은 물론, 製織裝置에 連結하여 이용할 수 있는 可能性을 모색하고자 하였다.

## II. 本 論

本 研究에서는 APPLE II형 퍼스날컴퓨터와 프린터를 사용하였으며, 칼라모니터 칼라프린터에 맞게 하였으나 흑백 프린터, 단색모니터에도 이용할 수 있게 하였으며 다음 두가지로 나누어 생각하였다.

첫째, dobby 직물조직 形成을 위한 프로그램.

둘째, 一完全組織과 經糸의 色糸配列, 緯糸의 色糸配列에 의한 混合된 織造무늬 形成을 위한 프로그램.

### 1. dobby 織物의 完全組織을 나타내는 프로그램의 作成

織物은 經糸와 緯糸의 交叉, 浮沈에 의하여 組織이 形成된다<sup>4)</sup>. 이것을 여러 개의 集團系統을 使用하는 dobby 織機에서 製織하는 경우 peg를 繫는 方法이나 系統通入方法등에 따라 織物組織이 달라진다. 紋樣이

### BASIC PATTERN (PEGGING PLAN)

```

      X X X X X X
      X X X X X X
    X X X X X X X
    X X X X X X X
      X X X X X X
      X X X X X X
  
```

### TABLET FORM (DRAFTING SYSTEM)

```

      X
    X X
  X X
X X
X X
X X
X X
  
```




Fig. 2. HONEYCOMB WEAVE.

심어져 있는 것을 經糸, 즉 上口開口로 하고 그와의 相應하는 通入圖의 표시점이 만나면 組織이 形成되도록 하였으며, 이것을 일완전조직도와 全體的인 組織무늬를 나타내고, 경, 위사의 굵기에 따라, 또한 무늬의 擴大效果를 볼 수 있게 하였으며 컴퓨터에 대한 專門知識이 없는 사람도 基本組織과 通經圖대로 key만 누르면 組織무늬가 즉시 모니터와 프린터에 出力되도록 Fig. 1과 같은 과정에 의해서 BASIC언어를 사용하여 프로그램을 作成하였다(Table 1).

위와 같은 프로그램으로 다음 事項을 順序대로 入力하면 完全組織무늬가 나타난다.

#### 1) 資料 入力

- 一完全組織圖의 經糸, 緯糸의 굵기
- 紋樣圖의 가로, 세로의 칸수
- 通入圖의 가로, 세로의 칸수
- 紋樣圖의 pattern
- 通入圖의 pattern

#### 2) 出 力

- 모니터에 完全組織圖를 나타냄
- 프린터에 完全組織圖를 나타냄

Table 1. 完全 組織圖를 나타내는 프로그램

```

100 REM -----
110 REM
120 REM   WOVEN FABRICS SIMULATION
130 REM       1985 , 6 , 9
140 REM
150 REM -----
160 REM
170 REM ( DATA INITIALIZATION )
180 DIM EPT$(30,30),TB$(30,30)
190 HOME : VTAB 3
200 INVERSE : PRINT " THREADS/LINE " ;: NORMAL : INPUT " ";N
210 EDZ = 0
220 X = 1:MX = 279:MY = 191
230 REM
240 REM (BRANCH TO EACH SUBROUTINES)
250 GOSUB 380
260 GOSUB 300
270 GOSUB 830
280 GOSUB 1070
290 GOTO 190
300 REM
310 REM (CLEAR THE SCREEN)
320 HGR2
330 HCOLOR= 3
340 FOR I = 0 TO 279
350 HPLOT I,0 TO I,191
360 NEXT I
370 RETURN
380 REM
390 REM (INPUT THE PARAMETERS)
400 HOME
410 VTAB 3
420 INVERSE : PRINT " PATTERN " ;: NORMAL : PRINT " SIZE (R,C) " ;: INPUT
    "" ;PF,PC
430 IR = PR
440 IC = PC
450 FG% = 1
460 GOSUB 590
470 IF I = 1 AND J = 1 THEN 400
480 HOME
490 VTAB 3
500 INVERSE : PRINT " COLUMN " ;: NORMAL : PRINT " OF TABLE IS " ;PC
510 INVERSE : PRINT " ROW " ;: NORMAL : PRINT " OF TABLE " ;: INPUT "" ;TR
520 TC = PC
530 IR = TR
540 IC = TC
550 FG% = 2
560 GOSUB 590
570 IF I = 1 AND J = 1 THEN 480
580 RETURN
590 VTAB 6: HTAB 2
600 FOR I = 1 TO IC
610 PRINT SPC( 2 - LEN ( STR$ ( I) ) ; I ;
620 NEXT I
630 PRINT
640 FOR I = 1 TO IR
650 PRINT I
660 NEXT I
670 PRINT : PRINT "(MARK TO PROPER POINT)".
680 FOR I = 1 TO IR
690 FOR J = 1 TO IC
700 VTAB 6 + I

```

```

710 HTAS 2 * J + 1
720 GET TP$
730 PRINT TP$
740 IF TP$ = CHR$(26) THEN PRINT CHR$(7): GOTO 660
750 IF TP$ = CHR$(27) AND I = 1 AND J = 1 THEN PRINT CHR$(7): GOTO
820
760 IF TP$ = CHR$(27) THEN PRINT CHR$(7): GOTO 660
770 IF TP$ < > "X" AND TP$ < > " " THEN PRINT CHR$(7): GOTO 700
780 IF FG% = 1 THEN EPT$(I,J) = TP$
790 IF FG% = 2 THEN TB$(I,J) = TP$
800 NEXT J
810 NEXT I
820 RETURN
830 REM
840 REM (DRAW PATTERNS)
850 HCOLOR= 4
860 FOR I = 1 TO PC
870 Y = MY - ((I - 1) * N)
880 FOR J = PR TO 1 STEP - 1
890 IF EPT$(J,I) < > "X" THEN 1010
900 CH = X + M + (TC - 1) * (N + 1)
910 IF CH > MX AND ED% = 1 THEN PRINT CHR$(7): GOTO 1050
920 IF CH > MX THEN X = 1:MY = MY - (PC * N): GOTO 860
930 FOR K = 1 TO TC
940 IF (TB$(J,K) < > "X") THEN 1000
950 D1 = X + (K - 1) * (N):D2 = D1 + (N - 1)
960 FOR L = N TO 1 STEP - 1
970 IF (Y - (N - 1)) < 0 THEN ED% = 1: GOTO 1030
980 HPLOT D1,Y - (N - L) TO D2,Y - (N - L)
990 NEXT L
1000 NEXT K
1010 NEXT J
1020 NEXT I
1030 X = X + (TC * N)
1040 GOTO 860
1050 RETURN
1060 REM
1070 REM (OUTPUT TO PRINTER)
1080 PR# 1
1090 PRINT "BASIC PATTERN"
1100 PRINT "-----"
1110 PRINT
1120 FOR I = 1 TO PR
1130 FOR J = 1 TO PC
1140 PRINT " ";EPT$(I,J);
1150 NEXT J: PRINT : NEXT I
1160 PRINT
1170 PRINT "TABLET FORM"
1180 PRINT "-----"
1190 FOR I = 1 TO TR
1200 FOR J = 1 TO TC
1210 PRINT " ";TB$(I,J);
1220 NEXT J: PRINT : NEXT I
1230 POKE 1913,34
1240 PRINT CHR$(17)
1250 PRINT : PR# 0
1260 PRINT : PRINT
1270 TEXT
1280 RETURN

```

BASIC PATTERN

```

X X      X X
 X X    X X
  X X  X X
 X   X X X

```

TABLET FORM

```

X           X
 X         X
  X       X
   X     X
    X   X
     X X

```




Fig. 3. WARP POINTED TWILL.

BASIC PATTERN

```

X X X X      X X X X
 X X X      X X X X
  X X      X X X X
 X      X X X X
        X X X X
         X X X X
          X X X X
           X X X X

```

TABLET FORM

```

X           X
 X         X
  X       X
   X     X
    X   X
     X X

```




Fig. 5. CURVED TWILL.

BASIC PATTERN

```

X X      X X X      X
 X      X X X      X X
  X X    X X X      X X X
 X X X  X X X      X X X
 X X X  X X X      X X X

```

TABLET FORM

```

X           X
 X         X
  X       X
   X     X
    X   X
     X X

```




Fig. 4. WARP BROKEN TWILL.

Fig. 2와 같은蜂巢組織을 예로 들면<sup>6,7)</sup>, 먼저 disk driver에 프로그램이 실린 disk를 넣으면 모니터에 紋控圖의 가로, 세로의 소요 칸수를 묻는다. 즉 PATTERN SIZE(R.C)가 나타나면 예를들어 7, 12를 입력

하면 MARK TO PROPER POINT라는 표시가 나오고 紋控圖 또는 자신이 원하는 任意的 紋控圖대로 pegging部分을 key X를 누르면 COLUME OF TABLE IS 12, ROW OF TABLE이 나오게 되고 7이라는 숫자를 key로 치면, Fig. 2와 같이 모니터의 스크린에 나타나고 곧이어 프린터에 찍혀 나온다. 같은 방법으로 하여 Fig. 3은 經山形斜紋織, Fig. 4는 經破斜紋織, Fig. 5는 曲斜紋織, Fig. 6은 組斜紋織, Fig. 7은 撚斜紋織을 컴퓨터의 printer에 나타나게 한 것이다. 실의 굵기 변화 및 조직무늬의 확대 무늬효과를 보고 싶으면 THREADS/LINE의 숫자를 1~10의 범위에서 큰 숫자로 바꿔줄수록 굵어진다. Fig. 7은 1로 한 것이고 Fig. 2~6은 2로한 경우이다.

여기서 紋控圖 또는 通入圖를 변경하면 任意的 組織을 얻을 수 있다.

## 2. 織物組織과 經·緯糸 색깔 및 配列에 의한 混 合무늬를 design하기 위한 프로그램 作成

製織에 의한 織物의 무늬 즉 織造무늬는 織物組織과 整經工程에서 定해진 經糸의 색깔과 配列數 및 配列方

BASIC PATTERN

```

X   X X   X
  X X   X X
X X   X X
X X   X X
  
```

TABLET FORM

```

      X   X
     X X X
    X X X
   X X X
  
```




Fig. 6. ENTWILING TWILL.

BASIC PATTERN

```

X X X           X X X X
X X           X X X X
X           X X X X
           X X X X
        X X X X
      X X X X
    X X X X
  X X X X
X X X X           X X X X
  
```

TABLET FORM

```

           X
        X X
      X X
    X X
  X X
X X
  
```




Fig. 7. CORKS CREW TWILL.

EPT S(15, 15), TB S(10, 10), CN S(8), CR%




Fig. 8. 직조무늬를 나타내는 과정

法和 織機의 복잡장치에 의한 緯糸의 색깔 및 配列順에 의한 混合된 효과가 무늬로서 나타난다. 그러니까 이것을 컴퓨터로 나타내려면 이 세가지 조건 즉, 완전 조직도, 경사배열, 위사배열을 입력시키면 된다. 경사, 위사의 색깔 표현은 [0] BLACK 1, [1] GREEN, [2] VIOLET, [3] WHITE 1, [4] BLACK 2, [5] ORANGE, [6] BLUE, [7] WHITE 2중에서 선택하여 그 숫자를 입력시키면 그 색깔을 나타내게 된다. 위 3가지 조건을 임의로 바꾸면 여러가지 변화있는 무늬를 design 할 수 있다.

Fig. 8과 같은 과정에 의해서 Table 2와 같은 프로그램을 作成하였다.

이상과 같은 프로그램에 의하여 다음과 같은 資料를 入力하면 즉시 무늬를 시각적으로 나타낸다.

- 1) 資料入力
  - 經·緯糸의 굵기
  - 경, 위사의 一完全 色糸配列의 所要칸 수
  - 一完全組織의 크기
  - 一完全 經, 緯糸의 色糸配列
  - 一完全組織圖
- 2) 出力

Table 2. 織造무늬 디자인 프로그램

```

100 REM -----
110 REM
120 REM   WOVEN FABRICS SIMULATION
130 REM       1985 . 6 . 8
140 REM
150 REM -----
160 REM
170 REM ( DATA INITIALIZATION )
180 DIM EPT$(15,15),TB$(12,12),CN$(8),CR%(36),CC%(36)
190 GOSUB 330: REM (COLOR NAMES)
200 HOME : VTAB 3
210 INVERSE : PRINT " THREADS/LINE " ;: NORMAL : INPUT " ";N
220 ED% = 0
230 X = 0:MX = 279:MY = 191
240 REM
250 REM (BRANCH TO EACH SUBROUTINES)
260 GOSUB 900: REM (INPUT COLORS)
270 GOSUB 540: REM (PARAMETERS)
280 GOSUB 400: REM (CLEAR SCREEN)
290 GOSUB 1260: REM (DRAW PATTERNS)
300 GOSUB 1490: REM (PRINT OUT)
310 GOTO 200: REM (REPEAT)
320 REM
330 REM (READ IN COLOR NAMES)
340 FOR I = 0 TO 7
350 READ CN$(I)
360 NEXT I
370 DATA "BLACK1","GREEN","VIOLET","WHITE1 "
380 DATA "BLACK2","ORANGE","BLUE ","WHITE2"
390 RETURN
400 REM
410 REM (CLEAR THE SCREEN)
420 HGR2
430 CP = CR
440 FOR I = MY TO N - 1 STEP - N
450 IF CR%(CP) = 4 THEN 510
460 HCOLOR= CR%(CP)
470 FOR J = I TO I - (N - 1) STEP - 1
480 IF J > 0 THEN Y = 0
490 HPLLOT 0,J TO MX,J
500 NEXT J
510 CP = CP - 1: IF CP < = 0 THEN CP = CR
520 NEXT I
530 RETURN
540 REM
550 REM (INPUT THE PARAMETERS)
560 HOME
570 VTAB 3
580 INVERSE : PRINT " PATTERN " ;: NORMAL : PRINT " SIZE (R,C) " ;: INPUT
" ";PR,PC
590 IR = PR
600 IC = PC
610 GOSUB 650
620 HOME
630 VTAB 3
640 RETURN
650 VTAB 6: HTAB 2
660 FOR I = 1 TO IC
670 PRINT SPC( 2 - LEN ( STR$ ( I ) ) ); I;

```

```

680 NEXT I
690 PRINT
700 FOR I = 1 TO IR
710 PRINT I
720 NEXT I
730 PRINT : PRINT "(MARK TO PROPER POINT)"
740 FOR I = 1 TO IR
750 FOR J = 1 TO IC
760 VTAB 6 + I
770 HTAB 2 * J + 1
780 GET TP$
790 PRINT TP$
800 IF I = 1 AND TP$ = CHR$ (27) AND J = 1 THEN PRINT CHR$ (7): GOTO
550
810 IF TP$ = CHR$ (27) THEN PRINT CHR$ (7): GOTO 740
820 IF TP$ = CHR$ (26) THEN PRINT CHR$ (7): GOTO 750
830 IF TP$ < > "X" AND TP$ < > " " THEN PRINT CHR$ (7): GOTO 760
840 EPT$(I,J) = TP$
850 NEXT J
860 NEXT I
870 RETURN
880 REM
890 REM
900 REM (INPUT THE COLORS)
910 REM
920 HOME
930 VTAB 3
940 INVERSE : PRINT " COLOR ";: NORMAL : PRINT " SIZE (R,C) ";: INPUT "
;CR,CC
950 HOME
960 FOR I = 0 TO 7
970 PRINT CHR$ (91);I; CHR$ (93);CN$(I): IF I < > 5 THEN PRINT " ";
980 NEXT I
990 PRINT
1000 VTAB 5: INVERSE
1010 FOR I = 1 TO CR
1020 PRINT I
1030 NEXT I
1040 PRINT : PRINT
1050 HTAB 4
1060 FOR I = 1 TO CC
1070 PRINT SPC( 2 - LEN ( STR$ (I)));I;
1080 NEXT I
1090 PRINT : NORMAL
1100 FOR I = CR TO 1 STEP - 1
1110 VTAB 5 + (I - 1): HTAB 4
1120 GET TP$: PRINT TP$
1130 IF (TP$ = CHR$ (27)) AND (I = CR) THEN PRINT CHR$ (7): GOTO 900
1140 IF TP$ = CHR$ (26) THEN PRINT CHR$ (7): GOTO 1100
1150 CR$(I) = INT ( VAL (TP$))
1160 NEXT I
1170 FOR I = 1 TO CC
1180 VTAB 5 + CR: HTAB 2 * I + 3
1190 GET TP$: PRINT TP$
1200 IF (TP$ = CHR$ (27)) AND (I = 1) THEN PRINT CHR$ (7): GOTO 1100

```



```

1210 IF TP$ = CHR$ (26) THEN PRINT CHR$ (7): GOTO 1170
1220 CC%(I) = INT ( VAL (TP$))
1230 NEXT I
1240 RETURN
1250 REM
1260 REM (DRAW PATTERNS)
1270 MN = 0
1280 FOR I = PR TO 1 STEP - 1
1290 Y = MY - ((PR - I) * N)
1300 FOR J = 1 TO PC
1310 IF EPT$(I,J) < > "X" THEN 1420
1320 CH = X + N + (PC - 1) * (N + 1)
1330 IF CH > MX AND ED% = 1 THEN PRINT CHR$ (7): GOTO 1430
1340 D1 = X + (J - 1) * N:ED = D1 + (N - 1)
1350 IF D2 > MX THEN D2 = MX
1360 IF D1 > MX THEN X = 0:MY = MY - (PC + N):MN = 0: GOTO 1130
1370 HCOLOR= CC%(J + MN)
1380 FOR L = N TO 1 STEP - 1
1390 IF (Y - (N - 1)) < 0 THEN ED% = 1: GOTO 1430
1400 HPLOT D1,Y - (N - L) TO D2,Y - (N - L)
1410 NEXT L
1420 NEXT J
1430 NEXT I
1440 IF MN < CC THEN MN = MN + PC
1450 IF MN > = CC THEN MN = 0
1460 X = X + (PC * N)
1470 GOTO 1280
1480 RETURN
1490 REM
1500 REM (OUTPUT TO PRINTER)
1510 PR# 1
1520 PRINT "THREADS/LINE ";N: PRINT
1530 PRINT "< BASIC PATTERN >"
1540 PRINT
1550 FOR I = 1 TO PP
1560 FOR J = 1 TO PC
1570 PRINT " ";EPT$(I,J);
1580 NEXT J: PRINT : NEXT I
1590 PRINT
1600 PRINT "-----"
1610 PRINT "      COLORS      "
1620 PRINT "-----"
1630 PRINT
1640 PRINT "ROW"
1650 PRINT "----"
1660 FOR I = CR TO 1 STEP - 1
1670 PRINT CR%(I);" ";
1680 NEXT I
1690 PRINT
1700 PRINT "COLUMN"
1710 PRINT "-----"
1720 FOR I = 1 TO CC
1730 PRINT CC%(I);" ";
1740 NEXT I
1750 PRINT
1760 POKE 1913,34
1770 PRINT CHR$ (17)
1780 TEXT
1790 PR# 0: PRINT : PRINT
1800 RETURN

```

THREADS/LINE 1

< BASIC PATTERN >  
(One complete design)

```

X X      X X
X X      X X
X      X X X      X
      X X      X X
X X      X X
X X      X X
X      X X      X
      X X      X X
    
```

-----  
COLORS  
-----

ROW (the order of colouring of wt.)  
-----

3 4 4 3 3 4 4 3

COLUMN (the order of colouring of wp.)  
-----

3 4 4 3 3 4 4 3




Fig. 9. CONTINUOUS LINE EFFECT.

- 一完全무늬를 모니터에 出力
- 一完全무늬를 프린터에 出力

Fig. 9와 같은 연속줄무늬 효과를 나타내는 것을 예로 들면 다음과 같다.

- THREADS/LINE(실의 굵기 표시)
- COLOR SIZE(R.C)(경사, 위사의 배열수) 8,16
- 色糸配列(흑백으로 나타내려면 3과 4선정) 경사 배열: 3443344, 위사 배열: 3443344
- PATTERN SIZE(R.C)(一完全組織圖의 가로, 세로의 크기): 8,8
- MARK TO PROPER POINT: 一完全組織圖의 經浮點대로 key X로 눌러주면 된다.

같은 방법에 의하여 Fig. 10은 shepherd's check 무늬, Fig. 11은 bird's eye 무늬, Fig. 12는 hairline

THREADS/LINE 1

< BASIC PATTERN >

```

X      X
      X X
X X
X
    
```

-----  
COLORS  
-----

ROW

-----  
3 3 4 4 4 4 3 3

COLUMN

-----  
3 3 4 4 4 4 3 3




Fig. 10. SHEPHERD'S CHECK PATTERNS.

THREADS/LINE 1

< BASIC PATTERN >

```

X X X X
X X X X
X X X X
      X X X X
X X X X X X X X
      X X X X
      X X X X
    
```

-----  
COLORS  
-----

ROW

-----  
3 4 4 3 3 4 4 3

COLUMN

-----  
3 4 4 3 3 4 4 3




Fig. 11. BIRD'S EYE AND SPOT EFFECTS.

THREADS/LINE 1  
 < BASIC PATTERN >

```

X X X
X X X
X X X
X X X
    
```

-----  
 COLORS  
 -----

ROW

```

4 4 3 4
COLUMN
-----
4 4 4 3
    
```




Fig. 12. HAIRLINE EFFECT.

THREADS/LINE 1  
 < BASIC PATTERN >

```

X X
X X
X X
X X
    
```

-----  
 COLORS  
 -----

ROW

```

4 3 4 3
COLUMN
-----
4 3 4 3
    
```




Fig. 13. STEP PATTERNS.

THREADS/LINE 1

< BASIC PATTERN >

```

X X
X X
X X
X X
    
```

-----  
 COLORS  
 -----

ROW

```

3 4 4 3
COLUMN
-----
3 4 4 3 3 4 4 3 3 3 4 4 3 3
    
```




Fig. 14. STRIPE EFFECTS.

THREADS/LINE 1

< BASIC PATTERN >

```

X X X X X X
X X X X X X
X X X X X X
X X X X X X
X X X X X X
X X X X X X
X X X X X X
X X X X X X
X X X X X X
X X X X X X
    
```

-----  
 COLORS  
 -----

ROW

```

3 4 4 3 3 4 4 3
COLUMN
-----
3 4 4 3 3 4 4 3
    
```




Fig. 15. STRIPE EFFECTS.

THREADS/LINE 1  
 < BASIC PATTERN >  
 X  
 X  
 -----  
 COLORS  
 -----  
 ROW  
 ---  
 4 3 4 3 4 4 3 4 3 4  
 COLUMN  
 -----  
 4 3 4 3 4 4 3 4 3 4




Fig. 16. CHECK EFFECT.

THREADS/LINE 1  
 BASIC PATTERN: >  
 X X X X  
 X X X X X  
 X X X X X  
 X X X X X  
 X X X X X  
 X X X X  
 X X X X  
 -----  
 COLORS  
 -----  
 ROW  
 ---  
 4 3 4 3 4 3 4 3  
 COLUMN  
 -----  
 3 4 3 4 3 4 3 4




Fig. 17. DERIVATIVE EFFECTS.

THREADS/LINE 3  
 < BASIC PATTERN >  
 X X X X  
 X X X X  
 X X X X  
 X X X X  
 X X X X  
 X X X X  
 X X X X  
 X X X X  
 -----  
 COLORS  
 -----  
 ROW  
 ---  
 4 3 4 3 4 3 4 3  
 COLUMN  
 -----  
 3 4 3 4 3 4 3 4




Fig. 18.

무늬, Fig. 13은 계단무늬, Fig. 14는 변화줄무늬, Fig. 15는 장식줄무늬, Fig. 16은 체크무늬, Fig. 17은 변화장식무늬, Fig. 18은 Fig. 17과 같으나 굵게 (3)한 것이다.

Ⅲ. 考 察

이상에서와 같이 퍼스날 컴퓨터를 利用하여 간단한 조작에 의하여 單位組織, 單位무늬의 표현은 물론 全體的인 무늬효과를 쉽게 시각적으로 표현할 수 있게 되었으며 이러한 研究를 계속 進進시키면 二重織, 多重織, 接結二重織 등의 組織도 간단히 나타내는 것이 可能하다는 것을 알 수 있으며 기억 용량이 큰 컴퓨터를 利用하여 高解像圖칼라모니터와 칼라프린터를 가진

컴퓨터를 사용하면 經緯糸의 굵기, 密度, 色相, 간격, 색사배열등을 입력하여 실제 제작한 직물의 外觀에 유사한 것의 표현할 수 있는 可能性을 發見할 수 있다. 또한 이러한 組織形成과 무늬디자인을 織機의 dobby 장치와 緯入 및 緯糸선택장치에 연결할 수 있는 하드웨어의 개발이 이루어져야 하겠다.

#### IV. 結 論

퍼스날컴퓨터를 利用하여 실제로 製織한 dobby 織物의 外觀과 같은 效果를 내어 織造무늬 디자인에 이를 응용하고자 검토한 결과.

1. 紋控圖와 通入圖를 入力시키면 즉시 正確한 完全組織圖와 全體效果를 시각적으로 나타낼 수 있으며,
2. 完全組織圖와 經糸의 色系配列, 緯糸의 色系配列을 入力시키면 色彩效果와 織造무늬를 나타내므로 무늬의 수정이 즉시 가능하게 됐으며,
3. 컴퓨터의 非專門人도 간단한 조작에 의하여 직물 무늬의 디자인에 컴퓨터를 이용할 수 있게 되었다.

#### 參 考 文 獻

- 1) 中島 勝, 長瀬英俊, 斗秀夫, パーソナル コンピュータによる織物の ばかし 柄の設計, 織機機械學會誌, 93~99, (1984).
- 2) Designing and weaving in Fancy weaving using microcomputers, ITS Fabric Forming, (1983).
- 3) Patten designing with microcomputers, ITS Fabric Forming, (1983).
- 4) 車瑞郁, APPLE II 基本 manual, 가남사, (1983).
- 5) WILLIAM WATSON, F.T.I., TEXTILE DESIGN AND COLOUR, LONGMANS, 20~62, (1954).
- 6) 禹範植, 織物組織學, 文運堂, 23~131, (1977).
- 7) 金澈中, APPLE II 퍼스날 컴퓨터 그래픽, 52~80, 光化門出版社, (1984).